CÓDIGO FUENTE DE “DISEÑO DE EJES v2.0”
Por: FROILÁN ANDRÉS MORALES SANABRIA <froilan1986@hotmail.com>
http://www.ticalc.org/archives/files/authors/95/9551.html
dis_ejes()

Local op1,op2,op3,op4,vs,ms,ts,s_ns,s_ys,k_ts,ns,u_f,u_e,u_m,ed

Try

 NewFold diseño_i

Else

 setFold(diseño_i)

EndTry

If getType(ingr_eje)≠"LIST"

 {2,2,3,2,"2078","22350","21000","25500","83000"}→ingr_eje

If dim(ingr_eje)=9 Then

 ingr_eje[1]→op1

 ingr_eje[2]→op2

 ingr_eje[3]→op3

 ingr_eje[4]→op4

 ingr_eje[5]→vs

 ingr_eje[6]→ms

 ingr_eje[7]→ts

 ingr_eje[8]→s_ns

 ingr_eje[9]→s_ys

EndIf

ClrIO

"Revise sus datos"→ed

Loop

 Dialog

  Title "DISEÑO DE EJES"

  DropDown "S. de Unidades… ",{"SI","Inglés"},op1

  DropDown "Tipo de carga… ",{"Cortante","Flexo-Torsión"},op2

  DropDown "D. Geométrica…",{"Cuñero Perfil","Cuñero Trineo","Chaflán Agudo","T.b.Redondeada","Ranura Anillo"},op3

  DropDown "Ver Fórmula……",{"No","Si"},op4

  Text ""

  Text "        Por: Froilán A. Morales S."

  Text "     <froilan1986@hotmail.com>"

 EndDlog

 If ok=0

  Exit

 If op4=2 Then

  If op2=1 Then

   Pause "D=√(2.94Kt*V*N/S'n)"

  Else

   Pause "D=((32N/π)√((Kt*M/S'n)^2+(3/4)(T/Sy)^2))^(1/3)"

  EndIf

 EndIf

 If op3=1

  "2.0"→k_ts

 If op3=2

  "1.6"→k_ts

 If op3=3

  "2.5"→k_ts

 If op3=4

  "1.5"→k_ts

 If op3=5

  "3.0"→k_ts

 "2.0"→ns

 If op1=1 Then

  "[N]"→u_f

  "[Pa]"→u_e

  "[N*m]"→u_m

 Else

  "[lb]"→u_f

  "[psi]"→u_e

  "[lb*in]"→u_m

 EndIf

 Loop

  If op2=1 Then

   Dialog

    Title "CORTANTE"

    Request "V "&u_f,vs,0

    Request "S'n "&u_e,s_ns

    Request "Kt",k_ts

    Request "N",ns

   EndDlog

   If ok=0

    Exit

   Try

    expr(vs)→v

   Else

    Pause ed

    Cycle

   EndTry

   vs→ingr_eje[5]

  Else

   Dialog

    Title "FLEXIÓN y/o TORSIÓN"

    Request "M "&u_m,ms,0

    Request "T "&u_m,ts

    Request "S'n "&u_e,s_ns

    Request "Sy "&u_e,s_ys

    Request "Kt",k_ts

    Request "N",ns

   EndDlog

   If ok=0

    Exit

   Try

    expr(ms)→m

    expr(ts)→t

    expr(s_ys)→s_y

   Else

    Pause ed

    Cycle

   EndTry

   ms→ingr_eje[6]

   ts→ingr_eje[7]

   s_ys→ingr_eje[9]

  EndIf

  Try

   expr(k_ts)→k_t

   expr(ns)→n

   expr(s_ns)→s_n

  Else

   Pause ed

   Cycle

  EndTry

  s_ns→ingr_eje[8]

  If op2=1 Then

   √(2.94*k_t*v*n/s_n)→d

  Else

   (32*n/(π)*√((k_t*m/s_n)^2+.75*(t/s_y)^2))^(1/3)→d

  EndIf

  If op1=1 Then

   Disp "D [m] ="

  Else

   Disp "D [in] ="

  EndIf

  If op3≠5 Then

   Pause d

  Else

   Disp d,"▶Incrementando en 6%"

   1.06*d→d

   Pause d

  EndIf

  Exit

 EndLoop

op1→ingr_eje[1]

op2→ingr_eje[2]

op3→ingr_eje[3]

op4→ingr_eje[4]

EndLoop

DispHome

EndPrgm

